REAL VALUE

CHAPTER 2 BASIC REAL OPTIONS

There are traded virtual (nearly) real options, which are very similar to traded financial options. For instance, on EUWAX, the boerse-stuttgart platform for multi-class and commodity derivatives, and on several bank platforms in London for covered warrants, there are traded options or warrants for crude oil, with a specified expiration date and exercise price. Since real options are usually embedded in investment opportunities, or existing real assets, the specification is not necessarily as precise as financial or commodity options, but still involve most of the factors governing those derivatives.

As with financial options, the value of real options depends on six variables:

- The value of the underlying asset;
- The exercise price, or investment cost;
- The time to maturity (expiration), sometimes infinite;
- The volatility in the value of the underlying asset (standard deviation), sometimes jumps;
- The risk free interest rate, during the life of the option; and
- The dividends paid on the underlying, or other "deemed" payouts during the project.

There are different types of real corporate options such as:

1 - The opportunity to "wait" and invest later (defer);

2 - The opportunity to abandon a project;

3 - The opportunity to scale back (contract) a project;

4 - The opportunity to make follow-up investments (expand);

5 - The opportunity for the firm to switch its output or production state or inputs;

6 - The opportunity to extend the life of a project;

7 - The opportunity to make phased investments (compound option);

8 - The opportunity to make investments where it is possible to choose the best of several assets (tenant choice and mixed uses).

One of the easiest formats for viewing real options is the binomial option pricing approach, where it is assumed that over time an asset value will move up or down, consistent with the expected volatility of the assets, and where the up followed by down movements “recombine”.

2.1 BINOMIAL OPTION PRICING MODEL[1]

The primary required assumptions for the one step binomial option pricing model are that the asset value V moves either up or down over the next period, there is a riskless interest rate, and the real option concern is either a call or put. Consider an asset with an initial valued V, volatility σ; a call option on the asset with initial price, C, or put option ,P; K= exercise price; and there is a possibility of borrowing or lending at the interest rate, r.

Probability Up = p = (a - d)/(u - d)

Probability Down = 1 - p

$a = e^{rt}$

$u = e^{\sigma \sqrt{t}}$ U= max[u*V-K,0] (2.1)

$d = e^{-\sigma \sqrt{t}} = 1/u$ D=max[d*V-K,0] (2.2)

t = time intervals as % of year

U’= max[K-u*V,0] (2.3)

D’=max[K-d*V,0] (2.4)

[1] Binomial is not a model but rather a methodology. A binomial with many steps can be used as a numerical solution for other option models. For a multiplicative n-period binomial process, the call value is

$$e^{-rt}\sum_{j=0}^{n} C_j^n p^j (1-p)^{n-j} MAX(u^j d^{n-j} V - K, 0)$$

$$C_j^n = \frac{n!}{j!\,(n-j)1}, \textit{the binomial coefficient}$$

Call option = e (-rt) *[p*(U)+(1-p)*D] (2.5)

Put option = e(-rt)*[p*(U')+(1-p)*D'] (2.6)

Suppose there is an asset worth 12.5 today with zero yield, which is expected to have a volatility of 40%, the riskless rate is 10% continuously compounded. You wish to value an opportunity to purchase the asset in one year's time, when the purchase price is 10% higher than today's purchase price of 10. Alternatively, you wish to value an opportunity to sell the asset in one year's time at 10% over today's purchase price of 10. The first opportunity is similar to a real call option, the second to a real put option.

Here is an illustration in Excel of simple one step call, and put, options.

Figure 2.1

	A	B	C	D
1	**One-Step Binomial Call**			
2	VOLATILITY	0.4		
3	RATE	0.1		
4	TIME	1		
5	up	1.492		
6	down	0.670		
7	probability	0.529		
8	up	EXP(B2*SQRT(B4))		
9	down	1/B5		
10	probability	(EXP(B3*B4)-B6)/(B5-B6)		
11	Time	0	1	
12				
13	Development		18.648	u
14	Values	12.5		
15			8.379	d
16				
17			11	u
18	Cost	10		
19			11	d
20				
21			7.648	u
22	Call Value	**3.663**		
23			0.000	d
24	C21=		MAX((C13-C17),0)	
25	C23=		MAX((C15-C19),0)	
26	B22=	EXP(-B3*B4)*(C21*B7+(1-B7)*C23)		

Figure 2.2

	A	B	C	D
1	**One-Step Binomial Put**			
2	VOLATILITY	0.4		
3	RATE	0.1		
4	TIME	1		
5	up	1.492		
6	down	0.670		
7	probability	0.529		
8	up	EXP(B2*SQRT(B4))		
9	down	1/B5		
10	probability	(EXP(B3*B4)-B6)/(B5-B6)		
11	Time	0	1	
12				
13	Development		18.648	u
14	Values	12.5		
15			8.379	d
16				
17			11	u
18	Cost	10		
19			11	d
20				
21			0.000	u
22	Put Value	**1.116**		
23			2.621	d
24	C21=		MAX((-C13+C17),0)	
25	C23=		MAX((-C15+C19),0)	
26	B22=	EXP(-B3*B4)*(C21*B7+(1-B7)*C23)		

The asset could be worth 18.6 in one year's time, so a call option to purchase the asset at 100% of today's purchase price of 10, is the present value of 18.6-11 times the probability of the upward movement. Alternatively, the asset could be worth 8.4 in one year's time, so a put option to sell the asset at 110% of today's purchase price of 10, is the present value of 11-8.4 times the probability of the downward movement.

Note that the primary differences between the one step binomial call and put option are cells C21 and C32. For a call, C21=MAX((C13-C19),0); for a put C23=MAX((-C15+C19),0).

2.1.1 CONSTRÓI CASE STUDY

CONSTRÓI, S.A., a real estate developer based in the north of Portugal, can start a housing development on a farm close to Almada. The farm is derelict, so that unless large investments are made in the near future, it will not be possible to carry on any type of farming on the site. Immediate development of the project would have a cost of €10 million (in present value terms). Gross present value of the development is €12.5 million. The volatility of real estate developments has been 40%. During the next year, it is expected prices will increase by u, or alternatively, prices might drop by approximately d. The risk free interest rate is 10% p.a.

CONSTROI's management team is worried by this uncertainty and is studying the possibility of delaying the beginning of the development, since if the market falls the company will not be able to recover the investment made. An eventual delay might result in higher costs. Development costs are expected to grow at 10% per annum. If the development does not start during the next year, it is expected that the area developed in the neighbourhood of the farm will become so dense, that the City Council will declare the farm site a public utility. The Council has already submitted a proposal to buy the land for €1.2 million at the end of the year.

Figure 2.3

	A	B	C	D	E	F	G	H
1	**PROPERTY DEVELOPMENT WITH DEFER & ABANDON OPTIONS**							
2	**One-Step Binomial Call & Put**							
3	Time	0	1		VOL	0.4		
4					RATE	0.1		
5	Development		18.648	u	TIME	1		
6	Values	12.5			u	1.492	EXP(F3*SQRT(F5))	
7			8.379	d	d	0.670	1/F6	
8					q	0.529	(EXP(F4*F5)-F7)/(F6-F7)	
9			11	u				
10	Cost	10						
11			11	d				
12								
13			7.648	u	C13=	C5-C9		
14	Defer Option	**3.663**			B14=	EXP(-F4*F5)*C13*F8		
15			0	d				
16		With Abandonment Option						
17	Combined		7.648	u				
18	Options	**4.174**			B18=	EXP(-F4*F5)*(C17*F8+(1-F8)*C19)		
19			1.200	d	C19=	1.200		

View this problem as a real call to invest in developing a site combined with a put to sell. CONSTRÓI's management team is trying to decide whether to start the development now or to wait until the situation in the housing market is clearer.

As shown in Figure 2.3, on the upside the real call is worth the present value of the upward price less the investment cost times the probability of an upward movement; on the downside the real put is worth the present value of the greater of the downward price less the investment cost, and the abandonment value times the probability of a downward movement. In the current situation, the best solution would be to wait before investing. The abandonment alternative is also valuable. The value of the option to defer is higher than the expected NPV of investing immediately (12.5-10.0=2.5). Consequently, the current value of the farm is the combined real call and put option value.

2.2 REAL EUROPEAN OPTIONS

Black and Scholes (1973) provided the analytical theory for a European real option, where the investment cost is considered constant, as might be the case in certain negotiated acquisitions, where the acquirer has a limited time to inspect the books and arrive at an investment decision at a pre-arranged price.

Assuming that the investment decision has a limited life, and the underlying acquisition value has certain specified distribution characteristics, the value W(V) is:

$$W(V) = Ve^{-qT} N(d_1) - Ke^{-rT} N(d_2) \qquad (2.7)$$

where V is the current value of the underlying asset, e^{-qT} is the continuously compounded asset yield discount factor, e^{-rT} is the riskless interest rate discount factor, where K=investment price, r= risk-free rate, T= time to irreversible decisions, σ = volatility of the underlying asset V, N( ) is the cumulative normal distribution function and

$$d_1 = \frac{\ln(\frac{V}{K}) + (r - q + .5\sigma^2)T}{\sigma\sqrt{T}} \tag{2.8}$$

$$d_2 = d_1 - \sigma\sqrt{T} \tag{2.9}$$

Figure 2.4



Figure 2.5



As usual in European options, the longer the time to the irreversible decision as shown in Figure 2.4, and the greater the volatility of the underlying acquisition, as shown in Figure 2.5, the more valuable the real call option.

The limitations of the real European option model are that it is only for a fixed maturity, as a European option can only be exercised at maturity, and the investment cost is fixed. Thus this model doesn't indicate the best time to investment, so it is not useful as an improvement over the NPV rule for capital budgeting. Yet it is easy to use, and so provides a rapid check on the reality of a real option approaches. Figure 2.6 shows Black-Scholes in Excel.

Figure 2.6

	A	B	C
1			BASIC BLACK-SCHOLES
2			
3	INPUT		
4	ASSET	100	
5	EXERCISE	100	
6	RISKLESS RATE	0.06	
7	ASSET YIELD	0.02	
8	VOLATILITY	0.5	
9	TIME / YEARS	1	
10			
11	OUTPUT		
12	CALL PRICE	20.95	B4*B17*EXP(-B7*B9)-B5*B18*EXP(-B6*B9)
13	CALL INTRINSIC	0.00	IF(B4>B5,B4-B5,0)
14			
15	d1	0.33	(LN(B4/B5)+(B6-B7+B8^2/2)*B9)/(B8*SQRT(B9))
16	d2	-0.17	B15-B8*SQRT(B9)
17	N(d1)	0.63	NORMSDIST(B15)
18	N(d2)	0.43	NORMSDIST(B16)

Here are two easy applications of the European option model.

2.2.1 DEFERRAL OPTION

A deferral option is normally a call option found in most projects where the owner holds the right to delay the date at which the project will start to be developed, where the Intrinsic Value + Time Premium = Option Value. The Time Premium equals the value of being able to wait.

Mini-case Example - Defer

A patent for a new toy can be bought. In order to develop the infrastructure needed to make and sell the product, it is necessary to invest £100. Under normal circumstances, the toy should be able to generate net profits of £20 a year for 6 years. Starting in 6 years, the demand for this type of product is highly uncertain, leading to an annual standard deviation of 45% in net profits. How much should one pay for the patent, if risky capital cost 12% and the risk free is 5%?

Figure 2.7 shows the solution in Excel.

Figure 2.7

	A	B
1	**DEFERRAL OPTION FOR PATENT ON NEW TOY**	
2	NCF Per Year	**£20**
3	ASSET VALUE	£82.23
4	INVESTMENT COST	£100
5	RISKLESS RATE	0.05
6	DISCOUNT RATE:RISKY ASSETS	0.12
7	TIME OF NCF	6
8	TIME OF INVESTMENT	1
9	VOLATILITY	0.45
10	CALL PRICE	**£10.14**
11		
12	d1	-0.10
13	d2	-0.55
14	N(d1)	0.461
15	N(d2)	0.292
16	ASSET VALUE	((1/B6)-(1/(B6*((1+B6)^B7))))*B2
17	ASSUME: DON'T HAVE TO DEVELOP INFRASTRUCTURE FOR ONE YEAR.	

The asset value is determined as an annuity of NCF for six years: $NCF[\frac{1}{r}-\frac{1}{r}(\frac{1}{(1+r_d)^6})]$.

If the investment cost is fixed, the maturity is exactly one year, and the volatility of the patent on the new toy is 45%, then the option to purchase the patent is £10.14, or around 12% of the total value of the asset before considering the investment cost. This is an out-of-the-money call option which has no current intrinsic value.

The main difficulties of a European option are the definition of the time horizon (given in this example) and the determination of the volatility (also given in this case). Typically, in real options the expected volatility is obtained by examining the volatility of other completed projects (toy patents in this case), or based on probabilities assigned to different market scenarios, or by calculating the historical volatility of the securities of other companies involved in similar businesses.

2.2.2 GROWTH OPTION

A growth or expansion real option is an option to expand or make a follow up investment, normally a call option giving the corresponding holder the right or ability to enter a market or scale up its operations by paying a certain (exercise) price.

Mini Case Example – Expansion

WMB, a well-known and successful carmaker is considering the possibility of buying BUCCANEER , a troubled company, widely known for producing vintage 4x4 vehicles. The price is £800 billion and the PV of the expected cash-flows is only £700 billion. However, by buying BUCCANEER, WMB is also acquiring technology that will enable an expansion to the 4x4 market segment, during the next 5 years. At the moment, this would imply an additional investment of £320 billion and the PV of the expected cash flows is £350 billion. The volatility of the estimate is 32%, and the riskfree rate is 6%. Would it make sense to buy BUCCANEER?

Figure 2.8 shows that the price of the acquisition is less than the current present value, so using the NPV rule this acquisition should be rejected. However, the growth real option value of the expansion option is substantial at the high volatility, and is £49 billion more than the loss on the net acquisition.

Figure 2.8

	A	B
1	**WMB EXPANSION OPTION**	
2		£ BILLIONS
3	ASSET VALUE	£350
4	INVESTMENT COST	£320
5	RISKLESS RATE	0.06
6	ASSET YIELD	0.00
7	TIME / YEARS	5
8	VOLATILITY	0.32
9		
10	CALL PRICE	148.92
11		
12	d1	0.90
13	d2	0.20
14	N(d1)	0.82
15	N(d2)	0.58
16		
17	PRICE	**£800.00**
18	CURRENT PRESENT VALUE	**£700.00**
19	BUCCANEER NET VALUE	**£48.92**
20	B19=	**-B17+B18+B10**

2.3 EUROPEAN EXCHANGE OPTION

Real exchange option are options where the investment cost (or other input) is stochastic along with the asset value (or output). Real switching options allow the holder to change between modes of operation. The easiest exchange or switching option model is Margrabe (1978), which simply substitutes a stochastic investment cost in the Black-Scholes model, and requires an exchange volatility rather than the volatility of the asset alone. A property development where the construction cost is variable is an exchange option.

Assume that the project present value is V, the cost of the investment in termination year is K, and that both are stochastic. A European exchange option has the value of:

$$F(V,K,\sigma_V,\sigma_K,\rho,\delta_V,\delta_K,t) = Ve^{-\delta_V t}N(d_1) - Ke^{-\delta_k t}N(d_2) \quad (2.10)$$

where t = time of the development, σ_V = instantaneous standard deviation of V, σ_K = instantaneous standard deviation of K, ρ = correlation between V and K, δ_V = income rate of developed system V, δ_K = yield of investment cost.

$$\sigma = \sqrt{\sigma_V^2 - 2\rho\sigma_V\sigma_K + \sigma_K^2} \tag{2.11}$$

N (.)=cumulative standard normal distribution function,

$$d_1 = \frac{\ln(V/K) + (\delta_K - \delta_V + 0.5\sigma^2)t}{\sigma\sqrt{t}} \qquad d_2 = d_1 - \sigma\sqrt{t}\ . \tag{2.12}$$

Figure 2.9

	A	B	C
1	**MARGRABE EXCHANGE**		
2			
3	BENEFIT	100.00	
4	COST	100.00	
5	TIME / YEARS	10.00	
6	RISKLESS RATE	0.04	
7	BENEFIT YIELD	0.04	
8	COST YIELD	0.04	
9	VOLATILITY BENEFIT	0.20	
10	VOLATILITY COST	0.20	
11	CORRELATION	**0.50**	
12	SPREAD VOLATILITY	0.20	SQRT(B9^2 + B10^2 -2*B11*B10*B9)
13			
14	SPREAD CALL OPTION VALUE	**16.64**	B3*B18*EXP(-B7*B5)-B4*B19*EXP(-B8*B5)
15			
16	d1	0.32	(LN(B3/B4)+(B8-B7+(B12^2)/2)*B5)/(B12*SQRT(B5))
17	d2	-0.32	B16-B12*SQRT(B5)
18	N(d1)	0.62	NORMSDIST(B16)
19	N(d2)	0.38	NORMSDIST(B17)

The exchange volatility happens to be the same as both the benefit and cost volatilities, due to the exact 50% correlation and equal volatilities, but this is somewhat unusual. This is an at-the-money exchange option which has no intrinsic call option value, but because the benefits might increase more than the costs, the opportunity to exchange future benefits for future costs is valuable.

SUMMARY

This chapter presents three simple real call option methods/models: binomial call & put, European call, and European exchange. In what circumstances can a real options valuation provide a final decision that is different from a traditional NPV approach? Real option value tends to be higher than NPV when uncertainty is high, since managers have flexibility to delay investing. If the NPV is close to zero, and there are limited immediate positive cash inflows, the time to defer a decision is valuable. Only arithmetic is required to calculate the results of a one step binomial model. The European option is appropriate where the option cannot be exercised until a specified time. The European exchange option is appropriate where the exercise price (investment cost) is uncertain.

EXERCISES

EXERCISE 2.1. Townbank has the option to require Canary Wharf to buy back a one year lease on 100,000 square feet (SF) of space at 33 Canada Square in 1 year at £42/SF. If current rents are £42/SF, Canary Wharf is a one step binomial world, rent volatility is 30% and the riskfree rate is 5%, what is the value of this option?

EXERCISE 2.2. To finance his education, Peter Carter wants to sell an option to Global Enterprises, whereby GE can acquire his post-MBS services for one year at £75,000. Currently Peter expects to be earning in a free market £70,000 post-MBA. He lives in a one step binomial world, his earnings volatility is 30% and the riskfree rate is 5%. What should he ask for this option?

EXERCISE 2.3. Citibank has a one year option to take up to an additional 100,000 square feet of space in Canary Wharf for one year at £45/SF. If Canary Wharf is a one step binomial world, current rent is £42/SF, rent volatility is 40% and the riskfree rate is 10%, what is the value of this option?

PROBLEMS

PROBLEM 2.4 An office building of 100,000 square feet in Manchester would be worth £500 per square foot, and costs £450 per square foot to build. MBA Build has received planning permission which expires in five years on a suitable plot of land which it wants to acquire for the development. The volatility of office buildings is 20%, interest rates 4%, and expected payout is 4%. What is the value of this land?

PROBLEM 2.5 Optimus holds a license to provide 3G in Portugal. The present value of expected net cash flows from operations is projected to be €1,200 million, the investment cost is €1,338 million, the riskless rate is 5%, the volatility of 3G revenues is 50%, and the time at which the investment is expected to be completed is two years. What is the value of this opportunity, if Optimus is alone in the market?

PROBLEM 2.6. Wendy Wang is given planning permission for a two year modular facility suitable for either retail or residential tenants that currently costs €1,700,000 to build. The current value of perpetual rent from retail tenants is worth €1,750,000 and from residential tenants €1,600,000. Each has a volatility of 20%, have an income rate of 4%, and the retail rent roll has a .5 correlation with the residential rent roll. The particular planning consent specifies that residential tenants may be substituted for retail tenants at the end of one year. What is this modular facility worth?